
*1 お詫びと訂正：某ふぁい氏の記事とはたぶんきっとおそらく関係がありません。

*2 Sigil：元々は魔術などの分野で使われる「紋章」といった意味を持つようです。カタカナだと

「シジル」になるらしい。なお、「プログラミング Perl vol.1」では"funny character"と書かれてい

るが、そう呼ばれているのを今知ったくらいに使われているところを見ない……。

use Perl::Object 0;
文 編集部 mitty

お詫びと訂正*1

前回の記事(16 号)の「目次」では、『次は第 1 章の「use による既存モジュールの活用と package

宣言」から始める』としていましたが、多角的に検討した結果、まずは初歩的な導入部分を用意

して、第 1 章は年度を改めてから始めた方が良かろう……という結論に至りました。従いまして

今回は第 0 章と称して、「Perl のデータ構造および CPAN からのモジュール導入」を扱います。

なお、掲載されているソースコードは、以下の処理系で確認しております。

l ActivePerl 5.12.2 build 1203 x86 および x64 on Windows 7 x64

l perl 5.10.1-8ubuntu2 x64 on Ubuntu 10.04.1 LTS (Lucid Lynx) x64

予定

第０章 Perl のデータ構造および CPAN からのモジュール導入 (今回)

第１章 use による既存モジュールの活用と package 宣言

第２章 Perl でのクラス・インスタンス・メソッド

第３章 既存モジュールの拡張・継承

第４章 POE: Perl Object Environment の紹介

第５章 POE を使ったマルチタスキング

第６章 POE によるイベントドリブン生活

第７章 並列クローラの作成

Perl における変数の型

Perl で書かれたコードに頻繁に出てくる「$」

「@」「%」などですが、これは Sigil
*2
と呼ばれ

ていて、変数の「型」を大まかに表していま

す。

「$」は「スカラー変数」、「@」は「配列変

数」、「%」は「連想配列変数(ハッシュ)」と呼

ばれ、右の「ソースコード 1」を見て貰うと

分かるかと思いますが、スカラー変数は値を

一つ、配列変数は値を複数持ちます。これ以

外にもファイルやディレクトリを扱う際に用

いる「ハンドル」や「サブルーチン」である

ことを表す「&」、Sigil の親玉である「型グロ

ブ」を示す「*」などがありますが、追々触れ

ていくことにしましょう。

use0-01.pl

1: #! /usr/bin/perl -w

2:

3: use strict;

4: use warnings;

5: use feature qw(say);

6:

7: my $hoge = 0;

8: my @hoge = (0, 1, 2, 3, 4, 5, 6, 7);

9: my %hoge = (

10: zero => 0,

11: one => 1,

12: two => 2,

13: );

14: say $hoge;

15: say join(", ", @hoge);

16: say "0 != 1" if ($hoge != $hoge[1]);

17: foreach my $key (keys %hoge) {

18: say "$key => $hoge{$key}";

19: }

ソースコード 1

*3 前回の記事：http://lab.mitty.jp/word/use_Perl_Object/use_Perl_Object.pdf あるいは、WORD16 号

「侵略されたいでゲソ号」をご覧下さい。

*4 配列変数：より正確には、値が複数代入できる変数を、単純な配列変数として扱うか連想配列

変数として扱うかによって、取り出される値が違うだけ過ぎません。

%hoge = (zero => 0, one => 1, two => 2); for (%hoge) { print $_, ", " };

# => one, 1, zero, 0, two, 2,

「%hoge」も上記で示されるように単純な配列として扱えます。ただし、「@hoge」とは異なり

(定義時や要素追加などの)順番は保存されません。

2

「ソースコード 1 の実行結果」で示されるように、「$hoge」と

「$hoge[1]」は違う変数です。そして「$hoge{one}」もまた違う

変数(の要素)を指しています。

なお、ここまでで疑問に思われた方も居るかと思いますが、Perl5

までの Perl にはいわゆる「int」「string」のような、データその

ものの種類を表す型はありません。

『例1)「==」と「eq」の違い』を見て下さい。[Ctrl+D]までの

太字の部分が入力したコード、[Ctrl+D]より下が出力結果です。

出力結果から分かるように、「==」演算子はオ

ペランドを数値として比較し、「eq」演算子は

文字列として比較します。文字列を数値とし

て評価すると「0」として扱われるため、「例1」

のような結果となります。

変数に型が無い、ということはその分記述

に曖昧さが生じますが、むしろ慣れるとこの

自由度の高さからストレス無く書けるように

なるでしょう。

なお、以降のソースコード中に出てくる「$_」

や「@_」などの、Perl 特有の特殊変数について

は、前回の記事*3 で一部触れているので、説明

を省略している箇所があります。

スカラーコンテキストとリストコンテキスト

Perl における変数の型として、「スカラー変数」と「配列変数*4」があることは既に述べました

が、Perl において「スカラー」および「リスト」の 2 単語は変数の型にとどまらず、「コンテキス

ト」と言われる文字通りコード中で式がどのように評価されるか、という意味でも使われます。

$ ./use0-01.pl

0

0, 1, 2, 3, 4, 5, 6, 7

0 != 1

one => 1

zero => 0

two => 2

$ perl -w

use feature qw(say);

say (("0" == 0) ? "true" : "false");

say (("0" == 0.0) ? "true" : "false");

say (("0" == "0.0") ? "true" : "false");

say (("0" eq 0) ? "true" : "false");

say (("0" eq "0.0") ? "true" : "false");

[Ctrl+D]

true

true

true

true

false

ソースコード 1 の実行結果

例 1)「==」と「eq」の違い

3

「ソースコード 2」の 9 行目を見て下さい。配列変

数である「@hoge」を、スカラー変数である「$fuga」

へ代入する場合、「@hoge」は「スカラーコンテキスト」

で評価されます。配列変数をスカラーコンテキストで

評価した場合、得られる値はリストの「要素数」とな

り、この場合「$fuga」には「8」が代入されます。

一方、10 行目では「@hoge」は「リストコンテキス

ト」で評価され、リストが返されます。

では、次の「例 2) スカラー変数とリストコンテキ

スト」はどうでしょうか。

「$hoge」自身はスカラー変数ですが、「()」で囲む

ことによってリストコンテキストとして扱われ、「@hog

e」の要素の一番目「0」が代入されます。

「コンテキスト」には他に「ブール値コンテキスト」

などがあり、これは真偽値として評価される場合が該

当します。「ブール値コンテキスト」は「スカラーコン

テキスト」の一種なので、変数に以下で述べる特定の

値以外が含まれている限り、真となります。

「ソースコード 3」におい

て「@hoge」から「pop」関数

によって一つずつ要素が取り

出されますが、要素がある限

り while ループからは抜け出

さないことになります。

Perl5 において「真」と

ならない、つまり「偽」と

して扱われるデータは以下

の通りです。

l 0

l 空文字列

l 空リスト

l undef (未定義値)

空ではない任意の文字列

は、数値として評価した場

合には「0」と評価される

ため、「偽」となります。

注意して欲しいのですが、「例 3) スカラー変数とリスト変数の真偽値」で示されるとおり、ブ

ール値コンテキストで評価すると「$hoge = 0」は「偽」となりますが、「@hoge = (0)」は要素が

一つあるため「真」となります。

use0-02.pl

1: #! /usr/bin/perl -w

2:

3: use strict;

4: use warnings;

5: use feature qw(say);

6:

7: my @hoge = (0, 1, 2, 3, 4, 5, 6, 7);

8:

9: my $fuga = @hoge;

10: my @fuga = @hoge;

11: say join(", ", $fuga);

12: say join(", ", @fuga);

$ ./use0-02.pl

8

0, 1, 2, 3, 4, 5, 6, 7

$ perl -w

($hoge) = @hoge = (0, 1, 2, 3, 4, 5, 6, 7); print $hoge;

[Ctrl+D]

0

use0-03.pl

1: #! /usr/bin/perl -w

2:

3: use strict;

4: use warnings;

5: use feature qw(say);

6:

7: my @hoge = (0, 1, 2, 3, 4, 5, 6, 7);

8:

9: while (@hoge) {

10: say pop @hoge;

11: }

$ ./use0-03.pl

7

6

5

4

3

2

1

0

ソースコード 2

ソースコード 2 の実行結果

例 2) スカラー変数とリストコンテキスト

ソースコード 3 実行結果

リファレンス

変数を受け取り値を 2 倍にして返す関数を考えます。受け取る変数がスカラー変数か配列変数

か分からないとすると、以下の「ソースコード 4」および「ソースコード 5」のように分けて書く

ことがまず考えられます。

しかし、扱いたいデータが常に、スカラー変数あるいは

配列変数のどちらかであるとは限りません。

ここで、「ソースコード 6」で示されるように「¥(円記号

あるいはバックスラッシュ)」演算子を用いて、引数となる変数の「リファレンス」を渡すように

すると、スカラー変数か配列変数かで関数を分ける必要が無くなります。

$ perl -w

@hoge = ($hoge = 0); print "true" if @hoge;

[Ctrl+D]

true

use0-04.pl

1: #! /usr/bin/perl -w

2:

3: use strict;

4: use warnings;

5: use feature qw(say);

6:

7: my $hoge = 1;

8:

9: say scalardouble($hoge);

10:

11: sub scalardouble {

12: my $scalar = shift;

13:

14: $scalar *= 2;

15: return $scalar;

16: }

use0-05.pl

1: #! /usr/bin/perl -w

2:

3: use strict;

4: use warnings;

5: use feature qw(say);

6:

7: my @hoge = (0, 1, 2, 3);

8:

9: say join(

10: ", ",

11: listdouble(@hoge)

12: );

13:

14: sub listdouble {

15: my @list = @_;

16:

17: return map {

18: $_ * 2

19: } @_;

20: }

$ ./use0-04.pl

2

$ ./use0-05.pl

0, 2, 4, 6

例3) スカラー変数とリスト変数の真偽値

ソースコード4 ソースコード5

ソースコード4の実行結果

ソースコード5の実行結果

WORD -xxx Edition- , 20xx

use Perl::Object 0;



5

また、いわゆる「n 次元配列」のような変数を扱い

たい場合を考えてみましょう。単純に考えると「例 4)

2 次元配列(失敗)」のようになりますが、これは単

純な一次元配列と同じになってしまいます。

実際に「$hoge[1][1]」に値を代入するには「例 5)

2 次元配列」のように書く必要があります。

「()」だった部分が「[]」に変わっていますが、

これにより「『配列のリファレンス』を要素に持つ配

列」が「@hoge」に代入されているのです。

さて、ここで「『ソースコード 6』では『${...}』

とか『@{...}』、あるいは『$$ret』といった変な Sigil

があるのに、リファレンスが含まれるはずの『例 5)』

ではそれが見あたらない」と気づいた方は鋭いです。

実は、「$hoge[1][1]」は「$hoge[1]->[1]」の省略

形で、リファレンスの中身を取り出す「デリファレ

ンス」演算子である「->」が隠れているのです。

use0-06.pl

1: #! /usr/bin/perl -w

2:

3: use strict;

4: use warnings;

5: use feature qw(say);

6:

7: my $hoge = 1;

8: my @hoge = (0, 1, 2, 3);

9:

10: my $ret = double(\$hoge);

11: say $$ret;

12: my @ret = double(\@hoge);

13: say join(

14: ", ",

15: @{ $ret[0] },

16: );

17:

18: sub double {

19: my $reference = shift;

20:

21: my $type = ref $reference;

22: if ($type eq "SCALAR") {

23: ${$reference} *= 2;

24: }

25: elsif ($type eq "ARRAY") {

26: @{$reference} =

27: map { $_ * 2 }

28: @{$reference};

29: }

30: else {

31: warn "something wrong";

32: }

33: return $reference;

34: }

$ ./use0-06.pl

2

0, 2, 4, 6

$ perl -w

use feature qw(say);

@hoge = ( (1, 2), (3, 4) );

say "no exist" unless exists $hoge[1][1];

say $hoge[3];

[Ctrl+D]

no exist

4

$ perl -w

@hoge = ( [1, 2], [3, 4] );

print $hoge[1][1];

[Ctrl+D]

4

ソースコード6例4) 2次元配列(失敗)

例5) 2次元配列

ソースコード6の実行結果

「例 6) 2 次元配列の構造」を見て頂くと、「@hoge」

の構造が理解頂けるかと思います。

つまり、「$hoge[1][1]」というコードは、ARRAY

のリファレンスである「$hoge[1]」を「->」演算子

でデリファレンスし、得られた配列の 2 番目の要素

「$hoge[1]->[1]」にアクセスしているのです。

「ソースコード 6」では「->」演算子ではなく、「$

{スカラーリファレンス}」や「@{配列リファレンス}」といった形でアクセスしているわけです。

なお、文法的に曖昧さが無い場合は「${...}」や「@{...}」の「{」と「}」は省略することが

可能で、「$$ret」のように書けます。

リファレンスによる複雑なデータ構造

「[]」による配列のリファレンス以外に、「{}」によるハッシュのリファレンス、「sub { ... }」

によるコードリファレンスなどがあります。これによって階層化された複雑なデータ構造が表現

できます。

「例 7)で示されるデータ構造が「$conf」に代入されている場合、「@{ $conf->{mail}{to} }」

にアクセスすると、「(info@example.com)」という配列が得られます。

$ perl -de 0

@hoge = ( [1, 2], [3, 4] );

print join(", ", @hoge);

[Ctrl+D]

ARRAY(0x11efdf0), ARRAY(0x129dce0)

例7) Twitter用botの設定ファイルをロードした後、Data::Dumperを通して出力したもの

{

'hashtag' => [

'#perlbot',

'#bot'

],

'allow' => {

'screen_name' => [

'twitter'

]

},

'mail' => {

'pickup' => [

'search',

'mention'

],

'to' => [

'info@example.com',

],

'subject' => 'new tweets for retweetbot are found',

'from' => 'retweetbot@twit.example.jp (Cron Daemon)',

'server' => 'localhost',

'contenttype' => 'text/plain; charset="ISO-2022-JP"'

},

}

例6) 2次元配列の構造

WORD -xxx Edition- , 20xx

use Perl::Object 0;



*5 規模が大きい：「89356 Modules, 8717 Uploaders」だそうです。(2011/01/21 08:55 JST)

*6 lwp-ua.pl：http://search.cpan.org/perldoc?LWP::UserAgent の「SYNOPSIS」ほぼそのままです。

CPAN

さて、ここまでで Perl のデータ構造

に関してはある程度理解して頂けたか

と思います。文法なども掘り下げると

色々と楽しいのですが、きりがないの

でこの辺にして、Perl でコーディング

するときに大変便利な CPAN について

解説しようと思います。

CPAN と は 「 Comprehensive Perl

Archive Network」の略語で、Ruby に

おける RubyGems や PHP における

PEAR に近いでしょうか。とはいうも

のの、あまりにも規模が大きい
*5
ため、

大抵のライブラリやちょっとしたツー

ルであれば車輪の再発明をするまでも

なく既に CPAN に登録されていること

が多いです。

CPAN に登録されているモジュール

は、 http://search.cpan.org/ から検索す

ることが出来ます。たとえば Twitter

に関する(と思われる)モジュールは

118 件(2011/01/21)あるようです。

CPAN モジュールの使い方

詳しくは今後の記事で再度触れますが、簡単に説明すると、「use Module::Name;」でモジュー

ルをロードし、「$instance = Module::Name->new( aruguments );」で「インスタンス変数」を新

しく用意して、そのインスタンス変数を通じてメソッド呼び出し、というのが基本的な使い方に

なります。モジュールによっては「クラスメソッド」にあたる関数が用意されていることもあり

ます。

独断と偏見で、よく使われるモジュールを紹介してみます。

・LWP::UserAgent

LibWWWPerl つまり WWW 関係の Perl ライブラリのうち、Perl をウェブクライアントとして使

う際に必要となる機能をまとめたモジュールです。wget コマンドをイメージするとどう使うか把

握しやすいかも知れません。

http://search.cpan.org/ に接続を試行し、成功すれば得られたコンテンツを、失敗した場合はステ

ータスを表示するコード「lwp-ua.pl*6」を次に示します。

WORD -xxx Edition- , 20xx

・Data::Dumper

「Data::Dumper 先生」と敬称で呼びたくなるほど便利なモジュールです。これを use しておく

と、既に出てきたように「Dumper」というクラスメソッドに引数を渡すことで、データ構造を再

現した形で中身を出力してくれます。print 文などに渡せば OK。

ウェブページをスクレイピングしたりする際に、メソッドから返ってくるオブジェクトの構造

がよく分からなかったりする場合は先生の出番です。 Twitter 関連のモジュールである

「 Net::Twitter::Lite」を使って RT ボットを作る過程で、返値を実際に出力したものが

http://lab.mitty.jp/trac/lab/wiki/Dev/Twitter/Perl/Dumper にありますので参考まで。

・Web::Scraper

最近その存在を知ったのでまだあまり使いこなせていませんが、非常に強力なスクレイピング

モジュールです。scraper というコマンドラインツールも付いてきます。同じような目的では、

「HTML::TreeBuilder」や「WWW::Mechanize」などといったモジュールもありますが、「Web::Scraper」

はそれらに比べ少ないコード量で済む設計になっているようです。

この他にも、本当に多種多様なモジュールがあります。是非自分で探してみて下さい。

CPAN モジュールの導入

「LWP::UserAgent」「Data::Dumper」は CPAN からダウンロードしなくても、perl 本体の配布パ

ッケージと一緒になって導入されているはずです。それ以外の「Web::Scraper」や「Net::Twitter」

などは自分で導入する必要があります。

このとき、導入方法には大まかに次の三つがあります。

lwp-ua.pl

1: #! /usr/bin/perl -w

2:

3: use LWP::UserAgent;

4:

5: my $ua = LWP::UserAgent->new;

6: $ua->timeout(10);

7: $ua->env_proxy;

8:

9: my $response = $ua->get('http://search.cpan.org/');

10:

11: if ($response->is_success) {

12: print $response->decoded_content; # or whatever

13: }

14: else {

15: die $response->status_line;

16: }

l CPAN 上のモジュールのページから tarball をダウンロードし、make する

l cpan コマンドにモジュール名を渡し、インストールする

l その他、OS 付属のパッケージ管理ツールを用いる

後に挙げた方法ほど容易になります。cpan コマンドに関してはウェブ上に文献がいくらでもあ

りますし、筆者自身普段はパッケージ管理ツールを用いているので、ここではパッケージ管理ツ

ールを用いた導入方法を簡単に説明しておきます。

・Windows

Windows 用のメジャーな Perl のバ

イナリは「ActivePerl」という名前で

配布されています。これには「ppm」

という、GUI ツールが付属していま

す。デフォルトのインストール先は

「C:¥Perl¥bin」です。

cpan コマンドなどもそうですが、

「ppm」もモジュール間の依存関係

を自動で解決して必要なモジュール

を追加で導入してくれます。また、

既に build 済みの形でダウンロード

されるため、cpan とは違い make を

行わず、(Perl プロジェクトではない)

外部のライブラリを必要とすること

がありません。

その代わり、CPAN に置かれてい

るパッケージの全てが網羅されてい

るわけではなく、特に GNU のライブラリなどに依存しているモジュールは無いことが多いです。

・Ubuntu

aptitude または apt-get で導入することが出来ます。パッケージ名にちょっと癖があり、たとえ

ば「WWW::Google::Calculator」であれば「aptitude install libwww-google-calculator-perl」で導入でき

ます。こちらは ppm よりも網羅されておらず、2000 パッケージ強しかないようです。

Perl のモジュールは依存関係が入れ子になっていることが多く、あるモジュールを導入する際

に、それが依存しているモジュールを導入しようとするとさらに依存が発生して……、と予想以

上にモジュールの導入に時間が掛かることが良くあります。このため、導入対象のモジュールそ

のものは ppm や aptitude などのパッケージ管理ツールには載っていない場合でも、あらかじめ依

存するモジュールを導入してから対象モジュールを cpan コマンドで導入すると、比較的楽に入れ

ることが出来るようです。

最後に

如何でしたでしょうか。かなり端折った部分も多く、まだまだ Perler として力不足を感じてい

ますが、この記事が楽しい Perl 生活のきっかけにでもなれば、幸いです。

*7 完全に不明：コード量ゼロですし。

10

ハイパー添削タイム(あるいは蛇足)

前回の記事に掲載しました mycat に対して、ありがたいご指摘と、添削があります。

l http://twitter.com/uasi/status/4133434205143040

今回の WORD に print for <>; って Perl のワンライナーが載ってたけど print <>; でもい

いんじゃないか。どっちもリストコンテキストだから print する前にファイル全体を読

み込むはず。それとも for <> は最適化されんのかな。

l http://twitter.com/uasi/status/4133886183350272

1 行ずつ読むなら print while <>; すべき。

while の場合山括弧演算子に

よって、ファイルから一行ず

つ読み込まれ画面に出力され

ますが、 for/foreach ではリス

トコンテキストで評価される

ため、ファイルから一度に全

て読み込んでしまいます。前

回の例では問題にはなりませ

んでしたが、読み込み対象が

巨大なファイルの場合は問題

が生じます。

右端のグラフは上段が CPU

使用率、下段がメモリ使用量

です。というか Windows のタ

スクマネージャーです。

①「mycat.pl」、②「mycat4.pl」の順番に、300MB のファイルを対象に読み込

みを実行しましたが、見ての通り for 文ではメモリが大量に確保されています。

ありがたいご指摘をして頂いた uasi 氏に感謝致します。

次に添削についてですが、「perl」のコマンドラインオプションに興味深い記述があります。

どうやら、ループを書かなくてもループを追加してくれるオプションがあるらしい……。

結果、mycat はさらに短くなりました。

ここまで来ると、何がしたいのかコードからは完全に不明*7 ですが、何事もなかったかのよう

に動作します。さすが Perl だ、なんともないぜ。

mycat.pl

1: #! /usr/bin/perl -w

2:

3: use strict;

4: use warnings;

5:

6: print for <>;

mycat4.pl

1: #! /usr/bin/perl -w

2:

3: use strict;

4: use warnings;

5:

6: print while <>;

$ perl --help

-n assume "while (<>) { ... }" loop around program

-p assume loop like -n but print line also, like sed

mycat5.pl

1: #! /usr/bin/perl -wp

*8 Artistic License：http://dev.perl.org/licenses/artistic.html

*9 2-clause BSD license：http://www.freebsd.org/copyright/freebsd-license.html

このコマンドラインオプションを用いることで、今回ソースコードに行番号を付けて出力する

ために用いた「add-line-number.pl」は、以下のように書くことが出来ます。

Perl で書かれたコードは大変シンプルになることがお分かり頂けたかと思います。

参考文献

l 「プログラミング Perl 第 3 版 VOLUME 1,2」(オライリー・ジャパン)

l 「PERL HACKS プロが教えるテクニック＆ツール 101 選」(オライリー・ジャパン)

l http://www.perl.org/

l http://www.cpan.org/

l perldoc コマンド

l Google 先生

免責

本記事の内容は、引用部分や素材など、著作権その他の権利が筆者に帰属しない物、あるいは

個別に但書きされている物を除き、Perl そのものと同じ「Artistic License*8」か、あるいは「2-clause

BSD license
*9
」に従って再利用できます。

また、記事タイトルの「Perl::Object」というモジュールは実際に存在しているわけではありま

せんのでご注意下さい。

本記事の内容は、以下の URL でも公開しております。

http://lab.mitty.jp/word/

内容に間違い・質問などありましたら、twitter:@mittyorz まで連絡して頂ければ幸いです。

add-line-number.pl

1: #! /usr/bin/perl -wp

2:

3: print "$.:\t";

